Q:

Solve system of equations with implicit function

I am trying to solve a system of 3 (non-linear) equations with two variables $x,\ y$ and one unknown function $f(x,y)$ of these two variables. Also, the system includes the change of this function $f$ upon the change of variables, i.e. something like:
$(1)\ \ \ \frac{1}{a+\frac{x}{f(x,y)}}(\frac{1}{f(x,y)} - \frac{x}{(f(x,y)^2}\frac{\partial f(x,y)}{\partial x}) + \frac{1}{-f(x,y)^2-\frac{x}{f(x,y)}}((\frac{x}{f(x,y)^2}-f(x,y))\frac{\partial f(x,y)}{\partial x} - \frac{1}{f(x,y)}) = 0$
$(2)\ \ \ \frac{1}{a+\frac{y}{f(x,y)}}(\frac{1}{f(x,y)} - \frac{y}{(f(x,y)^2}\frac{\partial f(x,y)}{\partial y}) + \frac{1}{-f(x,y)^2-\frac{y}{f(x,y)}}((\frac{y}{f(x,y)^2}-f(x,y))\frac{\partial f(x,y)}{\partial y} - \frac{1}{f(x,y)}) = 0$
$(3)\ \ \ \frac{1}{b^2f(x,y)^2-\frac{x}{bf(x,y)}}(b^2f(x,y)+\frac{xb}{(bf(x,y))^2}) + \frac{1}{c^2f(x,y)^2-\frac{y}{cf(x,y)}}(c^2f(x,y)+\frac{yc}{(cf(x,y))^2}) = 0$
I want to find $x,\ y,\ f(x,y)$. I have tried working with Solve but I had problems with passing the $f$ function both as an argument and object of interest. For example (a slightly simplified example):
Solve[{D[f[x, y], x] + a x + b y + f[x, y] == 1, 
  x - y - D[f[x, y], y] == 2, f[x, y] - x^2 - y^2 = 0}, {x, y}]
does not work. I have also tried with DSolve.
Could someone help me how to deal with this in mathematica?
Thanks in advance!

A:

The Question contains two different sets of equations.  I chose to solve the one in Mathematica format.  Begin with the third equation, which (after correcting a syntax error) is solved by
First@Solve[f[x, y] - x^2 - y^2 == 0, f[x, y]]
(* {f[x, y] -> x^2 + y^2} *)

Substitute this solution into the first and second equations, holding them Unevaluated so that the substitution occurs before the differentiation.
Unevaluated[{D[f[x, y], x] + a x + b y + f[x, y] == 1, 
   x - y - D[f[x, y], y] == 2}] /. %
(* {2 x + a x + x^2 + b y + y^2 == 1, x - 3 y == 2} *)

Solve[%, {x, y}]
(* {{x -> 1/20 (-14 - 9 a - 3 b - 3 Sqrt[44 + 28 a + 9 a^2 + 36 b + 6 a b + b^2]), 
     y -> 1/20 (-18 - 3 a - b - Sqrt[44 + 28 a + 9 a^2 + 36 b + 6 a b + b^2])}, 
    {x -> 1/20 (-14 - 9 a - 3 b + 3 Sqrt[44 + 28 a + 9 a^2 + 36 b + 6 a b + b^2]), 
     y -> 1/20 (-18 - 3 a - b + Sqrt[44 + 28 a + 9 a^2 + 36 b + 6 a b + b^2])}} *)

The more complex equation (not in Mathematica format) would be solved in the same way.  First, Solve the third equation for f.  Then, substitute the result into the first and second equations and Solve for x and y.

